# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=chafa
pkgver=1.12.4
pkgrel=3
pkgdesc="Image-to-text converter supporting a wide range of symbols and palettes, transparency, animations, etc."
arch=("loong64" "x86_64")
url="https://hpjansson.org/chafa/"
depends=('imagemagick' 'librsvg' 'libwebp' 'libxslt')
makedepends=('gtk-doc')
provides=('libchafa.so')
license=("LGPL")
source=(https://github.com/hpjansson/chafa/releases/download/$pkgver/chafa-$pkgver.tar.xz{,.asc})
sha512sums=('2732ff380fc6d61124298ae3d9958d676fe6ca2887ee11df057d4efa6f4ae9c074a7e1920a6a3db1ffb21882432769967f4453d52a0344e581993984133a0e9b'
            'SKIP')
validpgpkeys=('C01EDE5BB0D91E26D003662EC76BB9FEEAD12EA7')  # Hans Petter Jansson

build() {
  cd $pkgname-$pkgver

  ./configure --prefix=/usr --enable-man
  make
}

package() {
  cd $pkgname-$pkgver

  make DESTDIR="$pkgdir" install
}
